NBS

Technical Note 159

A FORTRAN CODE FOR CALCULATION OF EIGENVALUES AND EIGENFUNCTIONS IN REAL POTENTIAL WELLS

R. S. CASWELL

U. S. DEPARTMENT OF COMMERCE
NATIONAL BUREAU OF STANDARDS

THE NATIONAL BUREAU OF STANDARDS

Functions and Activities

The functions of the National Bureau of Standards are set forth in the Act of Congress, March 3, 1901, as amended by Congress in Public Law 619, 1950. These include the development and maintenance of the national standards of measurement and the provision of means and methods for making measurements consistent with these standards; the determination of physical constants and properties of materials; the development of methods and instruments for testing materials, devices, and structures; advisory services to government agencies on scientific and technical problems; invention and development of devices to serve special needs of the Government; and the development of standard practices, codes, and specifications. The work includes basic and applied research, development, engineering, instrumentation, testing, evaluation, calibration services, and various consultation and information services. Research projects are also performed for other government agencies when the work relates to and supplements the basic program of the Bureau or when the Bureau's unique competence is required. The scope of activities is suggested by the listing of divisions and sections on the inside of the back cover.

Publications

The results of the Bureau's research are published either in the Bureau's own series of publications or in the journals of professional and scientific societies. The Bureau itself publishes three periodicals available from the Government Printing Office: The Journal of Research, published in four separate sections, presents complete scientific and technical papers; the Technical News Bulletin presents summary and preliminary reports on work in progress; and Basic Radio Propagation Predictions provides data for determining the best frequencies to use for radio communications throughout the world. There are also five series of nonperiodical publications: Monographs, Applied Mathematics Series, Handbooks, Miscellaneous Publications, and Technical Notes.

A complete listing of the Bureau's publications can be found in National Bureau of Standards Circular 460, Publications of the National Bureau of Standards, 1901 to June 1947 ($1.25), and the Supplement to National Bureau of Standards Circular 460, July 1947 to June 1957 ($1.50), and Miscellaneous Publication 240, July 1957 to June 1960 (Includes Titles of Papers Published in Outside Journals 1950 to 1959) ($2.25); available from the Superintendent of Documents, Government Printing Office, Washington 25, D. C.

NATIONAL BUREAU OF STANDARDS

Technical Note

159

AUGUST 1962

A FORTRAN CODE FOR CALCULATION OF EIGENVALUES AND EIGENFUNCTIONS IN REAL POTENTIAL WELLS

R. S. Caswell

NBS Technical Notes are designed to supplement the Bureau's regular publications program. They provide a means for making available scientific data that are of transient or limited interest. Technical Notes may be listed or referred to in the open literature.

For sale by the Superintendent of Documents, U.S. Government Printing Office
Washington 25, D.C. - Price 25 cents

Contents

A Fortran Code for Calculation of Eigenvalues and Eigenfunctions in Real Potential Wells

Randall S. Caswell

A Fortran code has been developed for the calculation of eigenvalues and eigenfunctions for neutrons in real potential wells. A systematic procedure is given for approximate location of the eigenvalue and an automatic search procedure to determine the exact location. The code may be used for either bound or scattering states. In the case of scattering states, the criterion for maximum scattering (90° phase shift) is used to determine the energy of the state. The eigenvalues are determined by matching the numerically calculated logarithmic derivative (f_ℓ) inside the nucleus to the appropriate analytical logarithmic derivative for the region outside the nucleus. In an alternate mode of operation, the outside value of f_ℓ may be set arbitrarily, and a match made to this value. Sample results for a Woods-Saxon well with spin-orbit coupling for the case of oxygen-16 are shown. The code is in Fortran and was written for an IBM 7090 computer.

1. Introduction

Nuclear shell model calculations have until now employed almost exclusively harmonic oscillator wave functions for the evaluation of radial integrals. This has been done both for bound states and for states which actually belong to the continuum. The bound states evidently can be very well represented by harmonic oscillator wave functions. Any errors due to the difference between the "actual" optical potential and the harmonic oscillator potential will in general be much smaller than the errors due to neglect of higher configurations. This is, however, not necessarily true for continuum states.

The bound state wave functions have quite a different character than the continuum wave functions. In a treatment of continuum states it is thus desirable to have something better to work with than the harmonic oscillator eigenfunctions. The present report is concerned with the

evaluation of eigenfunctions specified by a given potential well and boundary conditions.

Calculations of eigenvalues in a diffuse potential have been reported by Ross, Mark, and Lawson (1956) where they used an analog computer and obtained eigenvalues accurate to about 0.1 Mev. The present calculation is for a digital computer (IBM 7090), is inherently capable of much higher accuracy, and also provides numerically integrated eigenfunctions as required.

2. Description of the Calculation

This code has been developed from a nuclear optical model code written for calculation of neutron elastic scattering which has been described elsewhere (Caswell, 1962). As a result some of the features are slightly more complicated than would be necessary for the present purpose. The main calculational steps are as follows: (1) input of data; (2) calculation of the real potential well, which is a Saxon well with spin-orbit coupling in the present code, but can easily be changed; (3) numerical integration of the wave functions, $u(r) = r\psi(r)$, out to a maximum radius which is taken where the nuclear potential has become negligibly small (except during the search procedure where smaller radii are used), and calculation of the logarithmic derivative, f_ℓ(inside); (4) calculation of the required f_ℓ(outside) if it is not fixed in the input data. The approximate location of the eigenvalues is determined by a survey versus energy of f_ℓ(inside), f_ℓ(outside), and the difference between them. An automatic search procedure finds the energy for which f_ℓ(inside) = f_ℓ(outside), and the wave function vs. radius is calculated and printed or punched on cards for the energy of the eigenvalue.

The potential is calculated by subroutine VR3 as a function of radius at up to 500 points. Spacing between points is determined by the variable HI up to radius RI, by HII up to RII, and by HIII up to RMAX. The potential used is:

$$V(r) = V_c \rho(r) - \alpha V_c \left(\frac{\hbar}{2Mc}\right)^2 (\vec{\ell}\cdot\vec{\sigma}) \frac{1}{r} \frac{d\rho}{dr} \qquad (1)$$

where $\rho(r) = \dfrac{1}{1 + \exp\left(\dfrac{r-R_o}{a}\right)}$, the "Saxon" potential; α is the strength of the spin-orbit interaction compared to the "Thomas" interaction for a nucleon; M is the nucleon mass; V_c is the central real potential; $\hbar/Mc$ is the Compton wave length for a nucleon; $\vec{\ell}$ is the orbital angular momentum of the neutron; $\vec{\sigma}$ is the Pauli spin operator of the incident neutron; R_o is the nuclear radius; and a is the "diffuseness" parameter of the potential. If $j = \ell + \frac{1}{2}$, (chosen by letting variable MSPIN = 1), then $\vec{\sigma}\cdot\vec{\ell} = \ell$, and the values of the variable VPLUS (potential for spin-orbit parallel) for different radii are calculated. If $j = \ell - \frac{1}{2}$, (chosen by letting MSPIN = -1), then $\vec{\sigma}\cdot\vec{\ell} = -(\ell+1)$, and the VMINUS values (potential for spin-orbit anti-parallel) are calculated. Other potential shapes could be used by simple changes in this subroutine. The four arbitrary parameters, VC, ALPHA, RADIUS, and A could be used to define the new potential with no changes outside of subroutine VR3. Any arbitrary potential could be read in point by point by replacing VR3 with an appropriate "READ" statement.

The calculation of f_ℓ(inside) and f_ℓ(outside), and the search procedure for matching the values is controlled by subroutine MATCH. First the radial part of the non-relativistic Schrödinger wave equation is

integrated out in radius for the appropriate ℓ:

$$\frac{d^2u_\ell(r)}{dr^2} + \frac{2m}{h^2}\left[E - \frac{\ell(\ell+1)h^2}{2mr^2} - V(r)\right]u_\ell(r) = 0. \qquad (2)$$

The values of the function and the first derivative at the first point (radius = HI) are determined from the relation $u_\ell(r) = C_\ell r^{\ell+1}$ as $r \to 0$, where C_ℓ is a constant. The second order Runge-Kutta method is used for the integration (see Zurmühl 1961). The integration is carried out by subroutine INTEG3 and is stopped when the radius exceeds RSTOP, and $f_\ell(\text{inside}) = R(du_\ell/dr)/u_\ell$ is calculated.

Calculation of f_ℓ(outside). The calculation may be run in two alternative modes. When MODE = 1, if the energy is negative (bound state), the outside wave function is required to have an f_ℓ which corresponds to an exponential function which asymptotically approaches zero for large r. If the energy is positive (scattering state), the criterion of a 90° phase shift is applied to the wave function at the radius RSTOP where the inside and outside functions must join smoothly. When MODE = 2, then f_ℓ is read in from the input data cards and no calculation of f_ℓ is made.

The outside wave function is the solution of equation (2) in the region of space where the nuclear potential V(r) = 0, but the centrifugal potential term is present. The solutions in general are spherical Bessel functions. For bound states, the "outside" or asymptotic solutions are spherical Hankel functions,

$$xh_\ell(x) = \left[xj_\ell(x) + ixn_\ell(x)\right] = F_\ell + iG_\ell \text{ where } x = i\sqrt{\frac{2m}{h^2}|E|}\, r.$$

For the 90° phase shift continuum case, the solutions are simply $xn_\ell(x) = G_\ell$. However for matching we require not the function but the

logarithmic derivative, f_ℓ. It is possible to calculate f_ℓ(outside) in a number of equivalent ways, for example by first evaluating the functions and then the derivatives from appropriate recursion relations. The calculation as done here uses the quotient functions of the Bessel functions discussed by Onoe (1958) for bound states, and a relation given by Blatt and Weisskopf (1952) for the continuum case. Details are given in Appendix 1.

3. Search Procedure

If the search for eigenvalues is carried out by matching the f_ℓ's at some distance outside the nuclear potential, particularly for states located deep in the potential well, there is a high probability of missing the state completely. The reason for this is that during the numerical integration of the wave function, as soon as the first radial point outside the nuclear potential at that energy is reached, the eigenfunction becomes in general a rapidly growing exponential, and f_ℓ becomes very large. Only when the energy is chosen extremely close to the correct eigenvalue does the wave function taper toward zero and have a small or negative value of f_ℓ. The quantities f_ℓ are thus very sensitive functions of the energy. In order to systematically find every eigenvalue, some procedure is needed to survey the eigenvalues under conditions of low sensitivity. The calculation was therefore split into two parts: first we perform a survey, and then we use an automatic search procedure starting with relatively small matching radius (which means low sensitivity) and increasing it stepwise to the desired value. This requires two separate runs on the computer, the first for the survey procedure to find the approximate eigenvalue location, and the second to find the exact location

by the automatic search procedure.

Survey procedure. When the variable NOMAX = 1, only one integration out in radius is made at each energy, and that integration is stopped at the edge of the well (for the particular energy) for negative energy, and at the value of RSTART for positive energy or for negative energy if RSTART corresponds to a smaller radius than the first criterion above. The maximum radii of integration for a typical situation are shown as the heavy black line in figure 1. The values of f_ℓ(inside), f_ℓ(outside) and the difference between them are printed out at the specified input energies taken typically 0.5 or 1 Mev apart. Typical results of this survey, called the "signature of the well" are shown in figures 2 and 3 for the case of $s_{\frac{1}{2}}$ and $p_{3/2}$ potential wells respectively. The small arrows indicate the accurate eigenvalue locations found by the automatic search procedure. The approximate eigenvalue locations corresponding to difference = 0, are used as starting energies for the automatic search procedure.

Automatic search procedure. First an integration of the wave function is carried out at the approximate eigenvalue energy. A second integration is carried out at a nearby energy. Using the differences (between f_ℓ's) found in these two trial calculations, a linear interpolation (or extrapolation) is made to an energy for which the difference is predicted to be zero. This is the predicted eigenvalue energy. Integration is carried out at the third energy, and the difference is again found. We now know three energies and the three corresponding differences. Based on this information, by quadratic interpolation, we predict the fourth energy for which the difference should be closer to zero. The calculation continues, energy prediction being made by quadratic interpolation, until



Fig. 1. Eigenvalues and the first three ℓ=0 eigenfunctions in a Wood-Saxon potential well. The parameters of the well are: V_c =-50 Mev, radius = 3.15 fermis, a = 0.650 fermis, spin-orbit term is 35 times the Thomas term. The shape of the well shown is for ℓ=0. During the survey procedure, the maximum radius is given by the heavy black line, the edge of the well for bound states, and 6.0 fermis for continuum states.



Fig. 2. Results of the survey procedure for the ℓ=0 well described in figure 1. Results are given for f_ℓ(inside), f_ℓ(outside) and for the difference between them.



Fig. 3. Results of the survey procedure for the ℓ=1, j = 3/2 well. Otherwise the presentation is as in figure 2.

either the difference is closer to zero than the difference specified in the input by the variable EPS, or until the maximum number of tries has been used up. The maximum number of tries is determined by the variable NOMAX.

The computer time for the sample automatic search procedure shown in Appendix 4, which required 21 repetitions of the integration of the wave function and matching procedure was 0.3 min. For a systematic search procedure at 69 energies, the computer time was 0.4 min.

The matching described above is carried out at a radius determined by the variable RSTART. RSTART should be chosen in the region where sensitivity of the f_ℓ changes with energy is low. Having determined the eigenvalue for a potential which is cut off at a value of radius determined by RSTART, we use this energy as the first trial energy for a larger value of r, which is determined by the variable RSTOP. The permissible size of a step in radius may be estimated from the WKB approximation. If we let $\varphi = f_\ell(\text{inside}) - f_\ell(\text{outside})$, then let $S(R) = \frac{\Delta\varphi}{\Delta E}$, a quantity which we have previously called the sensitivity. In the outward steps in radius, the increase in sensitivity, $\frac{S(R_{n+1})}{S(R_n)} = \sigma_r$, should be kept about constant for each increase in RSTOP. Approximately we may obtain σ_R as follows:

$$\ell n\, \sigma_R = \int_{R_n}^{R_{n+1}} \frac{1}{\hbar} \sqrt{2m(V-E)}\, dr$$

$$\approx \int_{R_n}^{R_{n+1}} \frac{1}{\hbar} \sqrt{2m|E|}\, dr \approx \sqrt{\frac{2m|E|}{\hbar}}\, \Delta R$$

since V soon becomes small in this region outside the well. The change in RSTOP is given by

$$\Delta R \approx \frac{\ell n \sigma_R}{\sqrt{2m|E|}/\hbar}$$

where σ_R is an arbitrary parameter determined by the input.

At the new matching radius, the first estimate of energy is the converged eigenvalue at the previous matching radius. The second estimate should be lower since VR^2 for the well will be slightly larger, causing the new eigenvalue to be slightly lower. At the second matching radius, the second trial eigenvalue is generated in the same way as at the first matching radius. At third and higher matching radii it is generated from the relation:

$$\Delta E_{n+1} = \frac{1}{\sigma_E} (\Delta E_n)$$

where σ_E is an arbitrary parameter determined from the input. Figure 4 shows a typical case of converging on an eigenvalue using this procedure of calculating at successive matching radii. The final match is carried out at the radius RMAX. When a converged eigenvalue is found at radius RMAX, the integration in radius is repeated and the wave function vs. radius is printed out. Also the value of the $\int u^2 dr$ is calculated by the trapezoidal rule for normalization purposes, and is printed out.

4. Sample Results

Sample results of the automatic search procedure for the case of Oxygen-16 are shown in figure 4. Eigenvalues are shown for the 1s, 2s, 3s, the two 1p states, the 1d state, and the eigenfunctions are shown for the three s states.

5. References

Amster, H. J. and Culpepper, L. M. (1957), Westinghouse Atomic Power Department Report WAPD-TM-87.

Blatt, J. M. and Weisskopf, V. F., (1952). Theoretical Nuclear Physics, P. 334, Wiley (New York).



Fig. 4. Example of the convergence on an eigenvalue. The convergence procedure is carried out for successive radii at 4.0 fermis, 5.25 fermis, and 6.40 fermis. Note the rapid increase in sensitivity (difference vs. energy) as the matching radius is increased.

Caswell, R. S. (1962). J. Research National Bureau of Standards, Section A, Physics and Chemistry, to be published.

Magnus, W. and Oberhettinger, F. (1954). Formulas and Theorems for the Functions of Mathematical Physics, p. 19, Chelsea (New York)

Onoe, Morio (1958). Tables of Modified Quotients of Bessel Functions of the First Kind for Real and Imaginary Arguments, Columbia University Press, New York.

Ross, A. A., Mark, Hans, and Lawson, R. D. (1956). Phys. Rev. 102, 1613.

Zurmühl, R. (1961). Praktische Mathematik für Ingenieure und Physiker, p. 412, Springer-Verlag (Berlin, Gottingen, Heidelberg).

The author would like to thank Dr. M. Danos for the suggestion of this problem and for helpful suggestions and discussions.

Appendix 1. Evaluation of f_ℓ(outside).

For bound states the logarithmic derivative, f_ℓ, may be obtained from one of the quotient functions of the Bessel functions which have been discussed by Onoe (1958).

$$f_\ell(\text{outside}) = R\,\frac{\left(\frac{du_\ell}{dr}\right)_R}{(u_\ell)_R} = R\,\frac{\left[\frac{d}{dr}\,(rh_\ell)\right]_R}{\left[r\,h_\ell\right]_R}$$

$$= X\,\frac{\left[\frac{d}{dx}\left(x\,h_\ell(x)\right)\right]_X}{\left[x\,h_\ell(x)\right]_X} \qquad (3)$$

where $X = i\sqrt{\frac{2m}{\hbar^2}|E|}\,R$.

but $\frac{d}{dx}\,(x\,h_\ell) = \frac{d}{dx}\left(x^{\ell+1}x^{-\ell}h_\ell(x)\right) = (\ell+1)\left(h_\ell(x)\right) - x\,h_{\ell+1}(x)$. (4)

Therefore,

$$f_\ell(\text{outside}) = X\,\frac{\left(\ell+1 - X\,h_{\ell+1}(X)\right)}{X\,h_\ell(X)} = \ell+1 - \frac{X\,h_{\ell+1}(X)}{h_\ell(X)} \qquad (5)$$

For a continuum state we have a similar relation in terms of the Neumann functions rather than the Hankel functions:

$$f_\ell = \ell+1 - \frac{X\,n_{\ell+1}(X)}{n_\ell(X)}\left(= X\,\frac{G_\ell'}{G_\ell}\right), \qquad (6)$$

where the prime refers to differentiation with respect to X. For a bound state,

$$f_\ell(\text{outside}) = \ell+1 - \tilde{\chi}^{(1)}_{\ell+\frac{1}{2}}(X),$$

in the notation of Onoe.

In the notation of Magnus and Oberhettinger (1954),

$\tilde{\chi}_\nu(1) = \frac{x\,K_{\nu+1}(x)}{K_\nu(x)}$. A recursion relation may be derived which

is used by the computer to calculate $\tilde{\chi}_v^{(1)}$ for the appropriate value of ℓ:

$$\tilde{\chi}_v^{(1)} = 2v + \frac{x^2}{\tilde{\chi}_{v-1}^{(1)}} \tag{7}$$

For continuum states, the criterion of a 90° phase shift, corresponding to a maximum value of the scattering cross section (or "resonance") is used to determine the location of the state. The relation for f_ℓ may be written as in equation (6) above

or $$f_\ell = \Delta_\ell + S_\ell \tan \xi_\ell, \tag{8}$$

where $\exp(-2i\xi_\ell) = \dfrac{G_\ell + i F_\ell}{G_\ell - i F_\ell}$; $\Delta_\ell = R\left[\dfrac{-G_\ell G_\ell' + F_\ell F_\ell'}{G_\ell^2 + F_\ell^2}\right]_{r=R}$;

and $S_\ell = R\left[\dfrac{G_\ell F_\ell' - F_\ell G_\ell'}{G_\ell^2 + F_\ell^2}\right]_{r=R}$. (See Blatt and Weisskopf, 1952).

Substituting for $\exp(-2i\xi_\ell)$ we have

$$f_\ell = \Delta_\ell - S_\ell \frac{2F_\ell G_\ell}{\left[(G_\ell^2 - F_\ell^2) + (F_\ell^2 + G_\ell^2)\right]}. \tag{9}$$

Equation (9) is equivalent to equation (5), and was the formula actually used in the calculation. Analytic expressions for all of the quantities in equation (9) appear in the report of Amster and Culpepper (1957).

Appendix 2. Meaning of the Input Variables

The meaning of the input variables are listed here for convenience in using the code. Format may be obtained from Appendix 3.

RADIUS	radius in the Saxon potential in fermis
A	diffuseness parameter of the Saxon potential in fermis
VC	central real potential in Mev (is negative for attractive potentials)
ALPHA	magnitude of the spin-orbit potential expressed in number of times larger than for the Thomas term of a nucleon
AMASS	atomic mass in a.m.u. of the nucleus except the single neutron whose states are being considered
CONV	parameter which determines energy spacing between first two trial eigenvalues, typical value 0.010
MDELTA	the radial potential is printed out at every MDELTA'th point
JPRINT	conditional print variable, = 1 for normal operation, = 3 maximum print out for code checking
L	orbital angular momentum
EPS	if the difference between f_ℓ's is less than EPS, then the eigenvalue is considered converged.
MSPIN	= 1 for spin-orbit parallel, = -1 for spin-orbit anti-parallel
HI	spacing between radial integration points at radii less than RI, in fermis
HII	same, but for radii between RI and RII
HIII	same, but for radii between RII and RMAX
RMAX	maximum radius, maximum value of RSTOP
NOMAX	maximum number of tries for convergence on an eigenvalue at a given value of RSTOP, = 1 for survey procedure, maximum value 20.
JPUNCH	= 1 for no punch out of wave function, = 2 for wave function to be punched on cards
KZ	number of trial energies for the given run, maximum = 100.

RSTART initial value of RSTOP

SIGMAR variable which indirectly determines spacing between successive values of radius (RSTOP) at which eigenvalue search is carried out, typically 2.0

SIGMAE variable which determines trial eigenvalue spacings after the first two trial radii, typically 10.0

MODE = 1 for normal search for bound and scattering states
= 2 for use with arbitrary f_ℓ determined from the input

EN(KY) trial energy in Mev (negative for bound states, positive for scattering states)

FLOUT(KY) value of arbitrary f_ℓ(outside)

Appendix 3. Format of Data Cards

First card: C in column 1, any desired information describing the run (such as name, date, element, etc.) in columns 7-72.

Second card: Variable	Ends in Column	Example	
RADIUS	8	3.150	(Note: decimal point location may be changed from that shown here, however print in the output will be as shown).
A	16	0.650	
VC	24	-50.000	
ALPHA	32	35.000	
AMASS	40	15.000	
CONV	48	0.010	
MDELTA	52	10	(Note that this and other variables without a decimal point are integers, decimal point may not be used).
JPRINT	56	1	
L	60	0	
EPS	70	0.0100000	
MSPIN	72	1	

Third card: Variable	Ends in Column	Example
HI	8	0.020
RI	16	0.100
HII	24	0.050
RII	32	1.000
HIII	40	0.010
RMAX	48	6.400
NOMAX	52	10
JPUNCH	56	2

Fourth card:

Variable	Ends in Column	Example
KZ	4	1
RSTART	12	4.0
SIGMAR	20	2.0
SIGMAE	28	10.0
MODE	32	1

Fifth and higher cards: If MODE = 1, list the energies with the last character successively in column 8, 16, 24, 32, 40, 48, 56, 64, 72, using as many cards as required for the KZ number of energies.

Example:

-29.000 -29.500

If MODE = 2, list the energies, one per card as follows:

Variable	Ends in Column	Example
EN	8	-29.000
FLOUT	24	0.2000000E+01

The variable FLOUT is in exponential notation and in the example means:

$0.2000000 \times 10^{+1} = 2.$

Appendix 4. Fortran Statements and Sample Output

```
C      EIGENVALUES,EIGENFUNCTIONS                    JULY 23,1962  05230
       READ 10
   10 FORMAT(72H
      1
       PRINT 10
       DIMENSION VPLUS(500),VMINUS(500),ENERGY(10,100),XP(1,100),
      1XMP(1,100),XPP(1,100),AJLP(1,100),AJLM(1,100),AILM(1,100),
      2AILP(1,100),FL(1,100),XX(100),EN(100),SPHTAN(10,100),XM(1,100),
      3 REFL(1,100),DIF(10 ),HTILDE(1,100),E(100),              PSI(4),
      4RRR(4),S(1,100),DELTA(1,100),TFG(1,100),FPG(1,100),GMF(1,100),
      5P(10,100),Q(10,100),FLOUT(100)
       COMMON VPLUS,VMINUS,XP,XM,XPP,XMP,AJLP,AJLM,AILP,AILM,FL,RR,X,V,
      1EN,FF,ELL,FRR,RADIUS,HI,A,E,VC,ALPHA,B,AMASS,EMAX,M,MDELTA,R,HII,
      2RII,HIII,RMAX,C,N,REXP,VCC,VS,AK,KZ,KX,KP,KY,K,XPLUS,XPLUSP,
      3XMINUS,XMINUP,AI,AII,AIII,CI,CII,CIII,H,XX,F,I,RI,MPRINT,JPRINT,
      4REFL,SPHTAN,KK,MM,CONV,NO,EPS,ENERGY,MSPIN,DIF,HTILDE,NOMAX,
      5JPUNCH,RSTOP,DELTA ,SIGMAE,SIGMAR,NTRY,DIFCNV,S,P,PSI,RRR,S,GMF,Q,
      6TFG,FPG,PMOM,MMM,RSTART,FLOUT,MODE,PROINT,L
       CALL INPUT3
       CALL VR3
       CALL MATCH
       CALL ENDJOB
        END
```

Note: Identical dimension and common statements must be included in every subroutine, but are omitted here for brevity.

```
      SUBROUTINE INPUT3
      READ 1,RADIUS,A,VC,ALPHA,AMASS,CONV,MDELTA,JPRINT,L ,EPS,MSPIN
      PRINT 4501
 4501 FORMAT(119H  RADIUS        A       VC  ALPHA     AMASS     CONV MDL JP
     1R   L     EPS    MSPIN                                                )
      PRINT1,RADIUS,A,VC,ALPHA,AMASS,CONV,MDELTA,JPRINT,L ,EPS,MSPIN
    1 FORMAT (6F8.3,3I4,1F10.7,1I2)
      KK=L+1
      READ 7,HI,RI,HII,RII,HIII,RMAX,NOMAX,JPUNCH
      PRINT 4502
 4502 FORMAT(119H      HI        RI      HII      RII     HIII     RMAX NOMAX
     1JPUNCH                                                               ),
      PRINT 7,HI,RI,HII,RII,HIII,RMAX,NOMAX,JPUNCH
    7 FORMAT(6F8.3,2I4)
      PRINT 4503
 4503 FORMAT(50H ENGS RSTART  SIGMAR  SIGMAE  MODE                          )
      READ  4040,KZ,RSTART,SIGMAR,SIGMAE,MODE
      PRINT 4040,KZ,RSTART,SIGMAR,SIGMAE,MODE
 4040 FORMAT(I4,3F8.1,I4)
      IF (MODE-1) 4081,4081,4082
 4081 PRINT 4504
 4504 FORMAT(50H TRIAL ENERGIES                                             )
      READ 4080,(EN(KY),KY=1,KZ)
      PRINT 4080,(EN(KY),KY=1,KZ)
 4080 FORMAT(9F8.3)
      GO TO 4085
 4082 PRINT 4083
 4083 FORMAT(50H  ENERGY        FL OUT                                      )
      READ  4084, (EN(KY),FLOUT(KY),KY=1,KZ)
      PRINT 4084, (EN(KY),FLOUT(KY),KY=1,KZ)
 4084 FORMAT (F8.3,E16.7)
 4085 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE VR3
      R=HI
      H=HI
      K=KK
      M=1
      PRINT 2091
2091  FORMAT(50H      VPLUS          VMINUS          R                    )
2090  DO 2210 I=1,1000
      I=I
    9 C= .0110270/A
   15 REXP=EXPF((R-RADIUS)/A)
2151  VCC=VC/(1.0+REXP)
2152  VS=ALPHA*VC*C*REXP/(((1.0+REXP)**2)*R)
   19 AK=K
      IF (MSPIN) 22,22,21
   21 VPLUS(I)=VCC+VS*(AK-1.0)
      GO TO 4020
   22 VMINUS(I)=VCC-VS*AK
      VMINUS(1)=VCC
4020  IF (M-MDELTA) 3150,28,28
   28 PRINT 30,     VPLUS(I),VMINUS(I),R
   30 FORMAT(2E15.7,F7.3)
      M=0
3150  M=M+1
2126  R=R+H
2130  IF(R-RMAX+.001) 2140,2220,2220
2140  IF(R-RII+.001) 2150,2160,2160
2150  IF(R-RI+.001) 2200,2180,2180
2160  H=HIII
      GO TO 2210
2180  H=HII
      GO TO 2210
2200  H=HI
2210  CONTINUE
2220  RETURN
       END
```

```
      SUBROUTINE INTEG3
    5 R=HI
      MMM=0
      K=KK
      AK=K
      IF (JPRINT-2) 35,35,4400
4400  PRINT 4401
4401  FORMAT(50H SUBROUTINE INTEG3 CALLED                              )
   35 XP(K,KY)=(HI**K)*1.0E-4
   37 XPP(K,KY)=AK*(HI**(K-1))*1.0E-4
   39 XM(K,KY)=XP(K,KY)
   41 XMP(K,KY)=XPP(K,KY)
   51 H=HI
   52 F=AMASS/(AMASS+1.00898)
   54 FF=F
      IF (MPRINT-1) 4090,4088,4090
4088  PRINT 4089
4089  FORMAT (50H  WAVE FUNCTION VS RADIUS                             )
      PROINT=0
      IF(JPUNCH-1) 4090,4090,4087
4087  PUNCH 4362
4362  FORMAT(50H  ENERGY                                               )
      PUNCH 4361, EN(KY)
4361  FORMAT(E13.5)
4090  DO 4210 I=1,1000
      I=I
      CALL KRR1
4126  R=R+H
4127  IF(NOMAX-1) 4130,4128,4130
4128  IF (R-0.2*RADIUS) 4130,4129,4129
4129  IF(EN(KY)-V) 4220,4130,4130
4130  IF(R-RSTOP+.001) 4140,4220,4220
4140  IF(R-RII+.001) 4150,4160,4160
4150  IF(R-RI+.001) 4200,4180,4180
4160  H=HIII
4170  GO TO 4210
4180  H=HII
4190  GO TO 4210
4200  H=HI
4210  CONTINUE
4220  IF (MSPIN) 4310,4230,4300
4300  PROINT=PROINT+(H/2.0)*XPLUS**2
      GO TO 4320
4310  PROINT=PROINT+(H/2.0)*XMINUS**2
4320  IF(MPRINT-1)4230,4330,4230
4330  PRINT 4340
4340  FORMAT(50H  NORMALIZATION                                        )
4350  PRINT 4360,PROINT
      IF(JPUNCH-1) 4230,4230,4351
4351  PUNCH 4340
      PUNCH 4360,PROINT
4360  FORMAT(1E15.7)
4230  RETURN
      END
```

```
5000 SUBROUTINE KRR1
     ELL=K-1
     IF (MSPIN) 5060,4125,5020
5020 XPLUS=XP(K,KY)
     PROINT=PROINT+H*XPLUS**2
5040 XPLUSP=XPP(K,KY)
  56 RR=R
  57 V=VPLUS(I)
  59 X=XPLUS
  61 CALL FR
  62 AI=FRR*H**2/2.0
  63 RR=R+H/2.0
  65 X=XPLUS+ XPLUSP*H/2.0+AI/4.0
  67 CALL FR
  69 AII =FRR*(H**2)/2.0
  70 RR=R+H
  71 X=XPLUS+XPLUSP*H+AII
  73 CALL FR
  75 AIII=FRR*H**2/2.0
 141 XPLUS=XPLUS+H*XPLUSP+(AI+2.0*AII)/3.0
 143 XPLUSP =XPLUSP +(AI +4.0*AII+AIII)/(3.0*H)
 180 XP(K,KY)=XPLUS
 184 XPP(K,KY)=XPLUSP
     GO TO 4110
5060 XMINUS=XM(K,KY)
     PROINT=PROINT+H*XMINUS**2
5080 XMINUP=XMP(K,KY)
  77 RR=R
  98 X=XMINUS
 100 V=VMINUS(I)
 101 CALL FR
 103 CI=FRR*H**2/2.0
 105 RR=R+H/2.0
 107 X=XMINUS+XMINUP*H/2.0+CI/4.0
 109 CALL FR
 111 CII=FRR*(H**2)/2.0
 112 RR=R+H
 113 X=XMINUS+XMINUP*H+CII
 115 CALL FR
 117 CIII =FRR*H**2/2.0
 149 XMINUS=XMINUS+H*XMINUP+(CI+2.0*CII)/3.0
 151 XMINUP= XMINUP +(CI+4.0*CII+CIII)/(3.0*H)
 188 XM(K,KY)=XMINUS
 192 XMP(K,KY)=XMINUP
4110 MMM=MMM+1
     IF(JPRINT-1) 4114,4114,4120
4120 IF(I-3) 4112,4112,4114
4112 PRINT 4113,XP(K,KY),XM(K,KY),XPP(K,KY),XMP(K,KY)
     PRINT 4113,XPLUS,XMINUS,XPLUSP,XMINUP
     PRINT 4113,AI,AII,AIII,CI,CII,CIII,H,R
     PRINT 4113,FRR,V,FF,X,RR
4114 IF(MPRINT-1) 4125,4115,4125
4115 IF (MSPIN) 4118,4118,4117
4117 PSI(MMM)=XP(K,KY)
     RRR(MMM)=RR
```

```
      GO TO 4119
4118  PSI(MMM)=XM(K,KY)
      RRR(MMM)=RR
4119  IF(MMM-4) 4125,4122,4125
4113  FORMAT (8E15.7)
4122  PRINT 4116,PSI(1),RRR(1),PSI(2),RRR(2),PSI(3),RRR(3),PSI(4),RRR(4)
4116  FORMAT (E13.5,F5.2,E13.5,F5.2,E13.5,F5.2,E13.5,F5.2)
      IF(JPUNCH-1)4135,4135,4132
4132  PUNCH 4116,PSI(1),RRR(1),PSI(2),RRR(2),PSI(3),RRR(3),PSI(4),RRR(4)
4135  MMM=0
4125  CONTINUE
      RETURN
      END



      SUBROUTINE FR
   44 FRR=-.04826*FF*(EN(KY)-20.721*ELL*(ELL+1.0)/(FF*RR**2)-V)*X
     . RETURN
      END
```

```
      SUBROUTINE MATCH
      DO 6200 KY=1,KZ
      MPRINT=2
      KY=KY
      RSTOP=RSTART
      CALL INTEG3
      PRINT 6180
 6180 FORMAT(119H     ENERGY        DIFFERENCE     FL INSIDE      FL OUT
     1SIDE  WAVEFNPLUS(R)  WAVEFNMINUS(R)      R                        )
      NTRY=1
      NO=1
  203 ENERGY(NO,KY)=EN(KY)
      IF (MSPIN) 208,208,204
  204 AJLP(K,KY)=XP(K,KY)*XPP(K,KY)
      AILP(K,KY)= XP(K,KY)**2
      REFL (K,KY)=R*AJLP(K,KY)/AILP(K,KY)
      GO TO 212
  208 AJLM(K,KY)=XM(K,KY)*XMP(K,KY)
  210 AILM(K,KY)= XM(K,KY)**2
      REFL (K,KY)=R*AJLM(K,KY)/AILM(K,KY)
  212 IF(MODE-1) 218,218,214
  214 DIF(NO)=REFL(KK,KY)-FLOUT(KY)
  218 IF(EN(KY)) 219,219,220
  219 PMOM=SQRTF(-.04826*F*EN(KY))
      IF (MODE-1) 2191,2191,6001
 2191 XX(KY)=R*PMOM
      HTILDE(1,KY)=1.0+XX(KY)
      SPHTAN(1,KY)=-XX(KY)
      DO 6000 K=2,KK
      AK=K
      HTILDE(K,KY)=(2.0*AK-1.0)+(XX(KY)**2)/HTILDE(K-1,KY)
 6000 SPHTAN(K,KY)=AK-HTILDE(K,KY)
      DIF(NO)=REFL(KK,KY)-SPHTAN(KK,KY)
      FLOUT(KY)=SPHTAN(KK,KY)
      GO TO 6001
  220 PMOM=    SQRTF(.04826*F*EN(KY))
      IF (MODE-1) 2201,2201,6001
 2201 XX(KY)=R*PMOM
  221 S(1,KY)=XX(KY)
      DELTA(1,KY)=0.0
  222 DO 226 K=1,KK
  223 AK=K
  224 S(K+1,KY)=(XX(KY)**2)*S(K,KY)/((AK-DELTA(K,KY))**2+S(K,KY)**2)
  226 DELTA(K+1,KY)=(AK-DELTA(K,KY))*(S(K+1,KY)/S(K,KY))-AK
  240 P(1,KY)=1.0
  244 P(2,KY)=1.0/XX(KY)
  246 Q(1,KY)=0.0
  248 Q(2,KY)=-1.0
  250 DO 254 K=3,KK
  251 AK=K
  252 P(K,KY)=(2.0*AK-3.0)*P(K-1,KY)/XX(KY)-P(K-2,KY)
  254 Q(K,KY)=(2.0*AK-3.0)*Q(K-1,KY)/XX(KY)-Q(K-2,KY)
  255 DO 268 K=1,KK
  256 GMF(K,KY)=(P(K,KY)**2-Q(K,KY)**2)*COSF(2.0*XX(KY))-2.0*P(K,KY)*
     1Q(K,KY)*SINF(2.0*XX(KY))
```

```
 258 TFG(K,KY)=2.0*P(K,KY)*Q(K,KY)*COSF(2.0*XX(KY))+(P(K,KY)**2
    1-Q(K,KY)**2)*SINF(2.0*XX(KY))
 268 FPG(K,KY)=P(K,KY)**2+Q(K,KY)**2
     FLOUT(KY)=DELTA(KK,KY)-S(KK,KY)*TFG(KK,KY)/(GMF(KK,KY)+FPG(KK,KY))
     DIF(NO)=REFL(KK,KY)-FLOUT(KY)
     IF(JPRINT-2) 6001,6001,271
 271 PRINT 272,XX(KY),S(KK,KY),DELTA(KK,KY),P(KK,KY),Q(KK,KY),
    1GMF(KK,KY),TFG(KK,KY),FPG(KK,KY)
 272 FORMAT(8E15.7)
6001 IF(NTRY-2) 6002,6002,6075
6002 IF(NO-NOMAX) 6004,6100,6100
6004 IF(NO-2) 6005,6080,6090
6005 IF(DIF(NO)+1.0)                          6010,6010,6020
6010 IF(DIF(NO)-1.0)                          6030,6040,6040
6020 EN(KY)=(1.0-CONV)*EN(KY)
     GO TO 6100
6030 EN(KY)=(1.0+CONV)*EN(KY)
     GO TO 6100
6040 IF(DIF(NO)+0.1)                        6050,6050,6060
6050 IF(DIF(NO)-0.1)                        6070,6100,6100
6060 EN(KY)=(1.0-0.2*CONV)*EN(KY)
     GO TO 6100
6070 EN(KY)=(1.0+0.2*CONV)*EN(KY)
     GO TO 6100
6075 IF(NO-2) 6076,6080,6090
6076 EN(KY)=(E(NTRY-1)-E(NTRY-2))/SIGMAE+E(NTRY-1)
     GO TO 6100
6080 EN(KY)=(ENERGY(NO-1,KY)*DIF(NO)-ENERGY(NO,KY)*DIF(NO-1))/
    1(DIF(NO)-DIF(NO-1))
     GO TO 6100
6090 EN(KY)=(((ENERGY(NO-2,KY)*DIF(NO-1)-ENERGY(NO-1,KY)*DIF(NO-2))*
    1DIF(NO)/(DIF(NO-1)-DIF(NO-2)))-((ENERGY(NO-1,KY)*DIF(NO)
    2-ENERGY(NO,KY)*DIF(NO-1))*DIF(NO-2)/(DIF(NO)-DIF(NO-1))))/
    3(DIF(NO)-DIF(NO-2))
6100 PRINT 6160,ENERGY(NO,KY),DIF(NO),REFL(KK,KY),FLOUT(KY),XP(KK,KY),
    1XM(KK,KY),R
6160 FORMAT(F15.7,5E15.7,F8.3)
     IF (NO-NOMAX) 6114,6200,6200
6114 IF(DIF (NO)+ EPS) 6120,6120,6115
6115 IF(DIF (NO)- EPS) 6130,6120,6120
6120 IF (JPRINT-1) 6123,6123,6121
6121 PRINT 6160, EN(KY)
6123 CALL INTEG3
     NO=NO+1
     GO TO 203
6130 E(NTRY)=ENERGY(NO,KY)
     IF(RSTOP -RMAX) 6131,6132,6132
6131 IF(JPRINT-2)6150,6132,6132
6132 MPRINT=1
6140 CALL INTEG3
6150 MPRINT=2
     IF (RSTOP-RMAX) 6165,6200,6200
6165 RSTOP=RSTOP+LOGF(SIGMAR)/PMOM
     IF (RSTOP-RMAX) 6168,6168,6167
6167 RSTOP=RMAX
6168 NTRY=NTRY+1
     CALL INTEG3
     NO=1
     GO TO 203
6200 CONTINUE
     RETURN
      END
```

```
  R S CASWELL  JUL 24,1962  OXYGEN SAXON WELL                    05230D32
 RADIUS        A       VC  ALPHA     AMASS     CONV MDL JPR   L     EPS    MSPIN
  3.150    0.650 -50.000  35.000   15.000    0.010  10   1    0 0.0100000 1
    HI        RI      HII      RII     HIII     RMAX NOMAX JPUNCH
  0.020    0.100    0.050    1.000    0.050    6.400  10   2
ENGS RSTART  SIGMAR  SIGMAE  MODE
  1     4.0     2.0    10.0   1
TRIAL ENERGIES
-29.000
    VPLUS          VMINUS          R
-0.4933571E 02 -0.0000000E-20  0.350
-0.4858813E 02 -0.0000000E-20  0.850
-0.4704951E 02 -0.0000000E-20  1.350
-0.4403985E 02 -0.0000000E-20  1.850
-0.3869766E 02 -0.0000000E-20  2.350
-0.3066896E 02 -0.0000000E-20  2.850
-0.2118392E 02 -0.0000000E-20  3.350
-0.1270445E 02 -0.0000000E-20  3.850
-0.6816266E 01 -0.0000000E-20  4.350
-0.3407750E 01 -0.0000000E-20  4.850
-0.1638992E 01 -0.0000000E-20  5.350
-0.7730581E 00 -0.0000000E-20  5.850
-0.3612086E-00 -0.0000000E-20  6.350
    ENERGY        DIFFERENCE     FL INSIDE      FL OUTSIDE  WAVEFNPLUS(R)  WAVEFNMINUS(R)      R
   -29.0000000 -0.4408910E 01 -0.8989446E 01 -0.4580536E 01  0.1670853E-04  0.2000000E-05   4.000
   -29.2899997 -0.1243811E 01 -0.5847193E 01 -0.4603382E 01  0.2255369E-04  0.2000000E-05   4.000
   -29.4039629 -0.3852737E-00 -0.4997603E 01 -0.4612329E 01  0.2489748E-04  0.2000000E-05   4.000
   -29.4600000 -0.1547027E-01 -0.4632192E 01 -0.4616722E 01  0.2605973E-04  0.2000000E-05   4.000
   -29.4624348 -0.8475780E-04 -0.4616997E 01 -0.4616913E 01  0.2611037E-04  0.2000000E-05   4.000
   -29.4624474 -0.2513295E 01 -0.7822746E 01 -0.5309450E 01  0.1148852E-04  0.2000000E-05   4.600
   -29.7570715  0.3369121E 01 -0.1966810E 01 -0.5335931E 01  0.2205862E-04  0.2000000E-05   4.600
   -29.5883269  0.9140120E 00 -0.4406769E 01 -0.5320781E 01  0.1595555E-04  0.2000000E-05   4.600
   -29.5422580 -0.9988016E-01 -0.5416517E 01 -0.5316637E 01  0.1431227E-04  0.2000000E-05   4.600
   -29.5461829 -0.4853845E-02 -0.5321844E 01 -0.5316990E 01  0.1445189E-04  0.2000000E-05   4.600
   -29.5463848 -0.1187312E 01 -0.7197843E 01 -0.6010530E 01  0.6834851E-05  0.2000000E-05   5.200
```

```
  -29.5545576 -0.2677180E-00 -0.6279080E 01 -0.6011362E 01  0.7368122E-05  0.2000000E-05   5.200
  -29.5569365 -0.2427983E-01 -0.6035884E 01 -0.6011604E 01  0.7523485E-05  0.2000000E-05   5.200
  -29.5571783 -0.1097918E-03 -0.6011738E 01 -0.6011628E 01  0.7539267E-05  0.2000000E-05   5.200
  -29.5571783 -0.5374961E 00 -0.7242773E 01 -0.6705277E 01  0.3681957E-05  0.2000000E-05   5.800
  -29.5582776 -0.4279304E-01 -0.6748195E 01 -0.6705402E 01  0.3819917E-05  0.2000000E-05   5.800
  -29.5583723 -0.1687109E-02 -0.6707100E 01 -0.6705413E 01  0.3831847E-05  0.2000000E-05   5.800
  -29.5583751 -0.2394826E-00 -0.7638559E 01 -0.7399076E 01  0.1897216E-05  0.2000000E-05   6.400
  -29.5584915 -0.1713628E-01 -0.7416227E 01 -0.7399091E 01  0.1925917E-05  0.2000000E-05   6.400
  -29.5585001 -0.1009881E-02 -0.7400102E 01 -0.7399092E 01  0.1928032E-05  0.2000000E-05   6.400
  -29.5584996 -0.1999021E-02 -0.7401091E 01 -0.7399092E 01  0.1927902E-05  0.2000000E-05   6.400
WAVE FUNCTION VS RADIUS
0.39995E-05 0.04  0.59976E-05 0.06  0.79935E-05 0.08  0.99865E-05 0.10
0.14952E-04 0.15  0.19883E-04 0.20  0.24770E-04 0.25  0.29601E-04 0.30
0.34365E-04 0.35  0.39053E-04 0.40  0.43653E-04 0.45  0.48156E-04 0.50
0.52553E-04 0.55  0.56833E-04 0.60  0.60988E-04 0.65  0.65009E-04 0.70
0.68887E-04 0.75  0.72616E-04 0.80  0.76187E-04 0.85  0.79594E-04 0.90
0.82830E-04 0.95  0.85890E-04 1.00  0.88768E-04 1.05  0.91459E-04 1.10
0.93959E-04 1.15  0.96265E-04 1.20  0.98373E-04 1.25  0.10028E-03 1.30
0.10199E-03 1.35  0.10349E-03 1.40  0.10479E-03 1.45  0.10589E-03 1.50
0.10678E-03 1.55  0.10748E-03 1.60  0.10797E-03 1.65  0.10827E-03 1.70
0.10838E-03 1.75  0.10830E-03 1.80  0.10803E-03 1.85  0.10758E-03 1.90
0.10696E-03 1.95  0.10618E-03 2.00  0.10523E-03 2.05  0.10413E-03 2.10
0.10288E-03 2.15  0.10149E-03 2.20  0.99978E-04 2.25  0.98342E-04 2.30
0.96592E-04 2.35  0.94740E-04 2.40  0.92793E-04 2.45  0.90761E-04 2.50
0.88654E-04 2.55  0.86481E-04 2.60  0.84250E-04 2.65  0.81972E-04 2.70
0.79654E-04 2.75  0.77307E-04 2.80  0.74938E-04 2.85  0.72556E-04 2.90
0.70169E-04 2.95  0.67783E-04 3.00  0.65407E-04 3.05  0.63047E-04 3.10
0.60709E-04 3.15  0.58399E-04 3.20  0.56123E-04 3.25  0.53884E-04 3.30
0.51688E-04 3.35  0.49537E-04 3.40  0.47437E-04 3.45  0.45388E-04 3.50
0.43394E-04 3.55  0.41458E-04 3.60  0.39579E-04 3.65  0.37760E-04 3.70
0.36001E-04 3.75  0.34303E-04 3.80  0.32666E-04 3.85  0.31090E-04 3.90
0.29574E-04 3.95  0.28119E-04 4.00  0.26722E-04 4.05  0.25383E-04 4.10
0.24102E-04 4.15  0.22876E-04 4.20  0.21704E-04 4.25  0.20585E-04 4.30
0.19517E-04 4.35  0.18499E-04 4.40  0.17529E-04 4.45  0.16606E-04 4.50
0.15727E-04 4.55  0.14891E-04 4.60  0.14096E-04 4.65  0.13341E-04 4.70
0.12624E-04 4.75  0.11944E-04 4.80  0.11298E-04 4.85  0.10685E-04 4.90
0.10104E-04 4.95  0.95538E-05 5.00  0.90320E-05 5.05  0.85378E-05 5.10
0.80697E-05 5.15  0.76264E-05 5.20  0.72069E-05 5.25  0.68098E-05 5.30
0.64341E-05 5.35  0.60786E-05 5.40  0.57424E-05 5.45  0.54244E-05 5.50
0.51237E-05 5.55  0.48393E-05 5.60  0.45706E-05 5.65  0.43165E-05 5.70
0.40764E-05 5.75  0.38494E-05 5.80  0.36350E-05 5.85  0.34324E-05 5.90
0.32409E-05 5.95  0.30601E-05 6.00  0.28892E-05 6.05  0.27278E-05 6.10
0.25754E-05 6.15  0.24314E-05 6.20  0.22954E-05 6.25  0.21669E-05 6.30
NORMALIZATION
0.2413705E-07
```

When the systematic search procedure is used, the printing format is the same as in the example. When the wave function is punched on cards, the format is as shown under "WAVE FUNCTION VS RADIUS," except that the energy is given first.

* U. S. GOVERNMENT PRINTING OFFICE: 1962 O - 656388

U. S. DEPARTMENT OF COMMERCE
Luther H. Hodges, *Secretary*

NATIONAL BUREAU OF STANDARDS
A. V. Astin, *Director*

THE NATIONAL BUREAU OF STANDARDS

The scope of activities of the National Bureau of Standards at its major laboratories in Washington, D.C., and Boulder, Colorado, is suggested in the following listing of the divisions and sections engaged in technical work. In general, each section carries out specialized research, development, and engineering in the field indicated by its title. A brief description of the activities, and of the resultant publications, appears on the inside of the front cover.

WASHINGTON, D. C.

Electricity. Resistance and Reactance. Electrochemistry. Electrical Instruments. Magnetic Measurements. Dielectrics. High Voltage.

Metrology. Photometry and Colorimetry. Refractometry. Photographic Research. Length. Engineering Metrology. Mass and Scale. Volumetry and Densimetry.

Heat. Temperature Physics. Heat Measurements. Cryogenic Physics. Equation of State. Statistical Physics.

Radiation Physics. X-ray. Radioactivity. Radiation Theory. High Energy Radiation. Radiological Equipment. Nucleonic Instrumentation. Neutron Physics.

Analytical and Inorganic Chemistry. Pure Substances. Spectrochemistry. Solution Chemistry. Standard Reference Materials. Applied Analytical Research. Crystal Chemistry.

Mechanics. Sound. Pressure and Vacuum. Fluid Mechanics. Engineering Mechanics. Rheology. Combustion Controls.

Polymers. Macromolecules: Synthesis and Structure. Polymer Chemistry. Polymer Physics. Polymer Characterization. Polymer Evaluation and Testing. Applied Polymer Standards and Research. Dental Research.

Metallurgy. Engineering Metallurgy. Microscopy and Diffraction. Metal Reactions. Metal Physics. Electrolysis and Metal Deposition.

Inorganic Solids. Engineering Ceramics. Glass. Solid State Chemistry. Crystal Growth. Physical Properties. Crystallography.

Building Research. Structural Engineering. Fire Research. Mechanical Systems. Organic Building Materials. Codes and Safety Standards. Heat Transfer. Inorganic Building Materials. Metallic Building Materials.

Applied Mathematics. Numerical Analysis. Computation. Statistical Engineering. Mathematical Physics. Operations Research.

Data Processing Systems. Components and Techniques. Computer Technology. Measurements Automation. Engineering Applications. Systems Analysis.

Atomic Physics. Spectroscopy. Infrared Spectroscopy. Solid State Physics. Electron Physics. Atomic Physics.

Instrumentation. Engineering Electronics. Electron Devices. Electronic Instrumentation. Mechanical Instruments. Basic Instrumentation.

Physical Chemistry. Thermochemistry. Surface Chemistry. Organic Chemistry. Molecular Spectroscopy. Molecular Kinetics. Mass Spectrometry.

Office of Weights and Measures.

BOULDER, COLO.

Cryogenic Engineering Laboratory. Cryogenic Equipment. Cryogenic Processes. Properties of Materials. Cryogenic Technical Services.

CENTRAL RADIO PROPAGATION LABORATORY

Ionosphere Research and Propagation. Low Frequency and Very Low Frequency Research. Ionosphere Research. Prediction Services. Sun-Earth Relationships. Field Engineering. Radio Warning Services. Vertical Soundings Research.

Radio Propagation Engineering. Data Reduction Instrumentation. Radio Noise. Tropospheric Measurements. Tropospheric Analysis. Propagation-Terrain Effects. Radio-Meteorology. Lower Atmosphere Physics.

Radio Systems. Applied Electromagnetic Theory. High Frequency and Very High Frequency Research. Modulation Research. Antenna Research. Navigation Systems.

Upper Atmosphere and Space Physics. Upper Atmosphere and Plasma Physics. Ionosphere and Exosphere Scatter. Airglow and Aurora. Ionospheric Radio Astronomy.

RADIO STANDARDS LABORATORY

Radio Physics. Radio Broadcast Service. Radio and Microwave Materials. Atomic Frequency and Time-Interval Standards. Millimeter-Wave Research.

Circuit Standards. High Frequency Electrical Standards. Microwave Circuit Standards. Electronic Calibration Center.


NBS